

Formal Verification Project Report

Solène Husseini

Liliane-Joy Dandy

January 9, 2023

Introduction

Static typing is a hallmark of functional programming languages. However, type checking and type inference are non-trivial tasks, thus the engines used to perform this task are often tightly integrated into compilers. In this project, we propose an approach of generating type constraints in Datalog, instead of computing them directly within the main compiler. This would ensure correctness and predictable performance of type checking, since typing rules could be implemented declaratively instead of through a conventional programming language. Our objective is to demonstrate the viability of this approach in a particular case by implementing a program that typechecks a simple polymorphic functional language, and where the user can define their own (maybe generic) types.

1 Background: a brief overview of Datalog

1.1 Language description

Datalog is a declarative language which has a few principal components:

- Facts, which define truths:

```
path(x, y).  
path(y, z).
```

Lowercase names are constant atoms.

- Rules, which derives new facts from known facts:

```
path(A, C) :- path(A, B), path(B, C).
```

Capitalised words are variables which get unified with facts. The comma operator denotes conjunction, and the fact to the left of `:-` is derived only if the right side unifies with a set of known facts.

- Queries, which try to find all known facts which satisfy it:

```
path(A, B)?  
> A=x, B=y  
> A=y, B=z  
> A=x, B=z // deduced by the rule
```

The goal of our project is therefore to convert the AST of our language into datalog facts, and provide datalog rules derived directly from typing rules to type all terms in the AST, and thus check whether it is well-typed.

As shown later, we also emit special rules while converting user defined types, breaking the clear separation of facts and rules, though we only emit facts when typechecking terms.

1.2 Evaluator

We need to use a Datalog evaluator to run the files that we generate. However, most stray a bit from the usual definition of the Datalog language, and introduce special features or different syntax. In order to get the project going, we settled for a simple datalog evaluator bundled with the Racket language. It is very limited though, both in features and performance. We tried to port the typechecker to Soufflé, a much more mature implementation, but it necessitated a modification of the compiler that we did not want to undertake. We still conducted a few tests and everything seemed to work fine with it.

2 Language presentation

We need a language to typecheck, thus we created a minimal functional programming language with Scala-like syntax. Types are defined as follows:

```
type Nat {  
  case z()  
  case s(Nat)  
}
```

It defines a type called `Nat`, with two constructors, `z` and `s`. There is no subtyping, and constructor arguments are not named, they can only be retrieved through a `match`.

Top-level terms are defined like this:

```
def two: Nat = { s(s(z)) }  
def add: Nat -> Nat -> Nat = {  
  \x: Nat. \y: Nat. x match { // no type inference, needs annotations  
    case z() => y  
    case s(x1) => s(add(x1, y))  
  }  
}  
// With sugar for top level functions:  
def add(x: Nat, y: Nat): Nat = { x match { ... } }  
  
def times4(x: Nat): Nat = {  
  let xtimes2: Nat = add(x, x) in  
  add(xtimes2, xtimes2)  
}  
// Desugared let into lambda:  
def times4(x: Nat): Nat = {  
  (\xtimes2: Nat. add(xtimes2, xtimes2))(add(x, x))  
}  
// Type parameters restricted to top level definitions  
def id[T](t: T): T = { t }  
  
// All together  
def map[T, U](f: T -> U, ls: List[T]): List[U] = {  
  ls match {  
    case nil() => nil()  
    case cons(h, t) => cons(f(h), map[T, U](f, t))  
  }  
}
```

It is worth pointing out that `let` is purely a syntactic sugar. We do not implement let polymorphism through a `let` construct, but instead restrict polymorphic definitions to top-level bindings.

Function application is also implicitly curried since functions are desugared to take only one argument, i.e. these are equivalent:

```
foo(x, y) == (foo(x))(y) // needs extra parentheses
```

We also did not implement an interpreter since it was not strictly necessary for typechecking, as well as the fact that it would have taken even more time to write.

2.1 Variable renaming

While translating the AST into datalog, we convert all of the function bodies together at once. Therefore, we need to avoid conflicts between variables with the same name appearing in different scopes. To that end, we tried two solutions:

- Inspired by [Pacak et al., 2020], we introduce a new relation `context_typed(E, C)` which would bind an expression `E` to a context `C`, where the context would bind variables to types. This allows different variables to refer to different types depending on their contexts (scopes).
- Originally done to simplify the compiler, we rename each variable with a program-global identifier. Then, no other precaution needs to be taken when converting the AST to Datalog since every identifier will be unique.

We ended up settling on the second solution after a while because it simplified a lot the Datalog facts and rules we would have had to emit.

3 Datalog translation of programs without generic types

We divided our work on typechecking in Datalog into two different parts: at first, we try to typecheck all programs that do not contain generic types and polymorphism. This has been fully automated. In a second stage, we add all the support for generic types and polymorphism. Note that we do not perform type inference.

3.1 Preliminary remark on expression names

When translating the inference rules or the syntax tree of the program, we will express a wide range of facts and relations. We emphasize that for every relation we define, of the form

$$\text{relation_name}(\text{arg_1}, \text{arg_2}, \dots, \text{arg_n})$$

the first argument is always the name of the expression, and the others are the relation arguments.

For instance, in the fact

```
app(app0, f, x).
```

`app0` is the name of the expression (here a function application), and `f` and `x` are the arguments. This means that we give this expression the name `app0 = f x`.

3.2 Generate data types as Datalog relations

As our language allows for user-defined types, these types must be exported to Datalog. We developed two approaches for data type generation.

The first one is much more elegant in theory but more complicated (and less "readable") in practice. It consists in defining the constructors as constants (when they do not take arguments) or functions. The `Nat` data type, defined in our language as

```
type Nat {  
  case z()  
  case succ(Nat)  
}
```

is translated into Datalog as

```
type(nat).  
typed(z,nat).  
fun(funNatNat,nat,nat).  
typed(succ,funNatNat).
```

Here, we give the name `funNatNat` to the type $Nat \rightarrow Nat$, we type the `succ` constructor as a function from `Nat` to `Nat`, and `z` as a constant of type `Nat`.

The main inconvenient with this approach concerns constructors with multiple arguments. As we desugar the multiple argument functions as a chain of partial applications with one argument, we find ourselves dealing with partial applications of constructors, which we prefer avoiding. This becomes even worse when tackling the challenge of generic types.

This is why we will generally take a second (more pragmatic) approach, which consists in defining the constructors as relations:

```
type(nat).  
typed(E,Nat) :- z(E).  
typed(E,Nat) :- succ(E,E1), typed(E1,Nat).
```

As usual, the first argument of `z` and `succ`, `E`, is the name of the expression. We can read the previous rules as follows:

- An expression `e` containing only the `z` constructor (`e = z`) is of type `Nat`.
- An expression `e = succ(e1)`, when `e1` is of type `Nat`, is of type `Nat`.

3.3 Inference rules

To typecheck programs correctly, we need to translate the inference rules to Datalog.

3.3.1 Lambdas

The inference rule for function definition is

$$\frac{C, v : T_v \vdash b : T_b}{C \vdash \lambda v : T_v. b : T_v \rightarrow T_b}$$

which can be translated into a Datalog rule

```
typed(X,T) :- lambda(X,V,B), typed(B,Tb), fun(T,Tv,Tb).
```

In this rule, X is the name of the complete expression ($x = \lambda v : T_v.b$), V is the variable, T_v the variable's type, and B the body of the function. This expression X is well typed and of type $T = T_v \rightarrow T_b$ if B is of type T_b .

Remark: Our language supports function with multiple arguments: they are desugared as a chain of partial applications with one argument each.

3.3.2 Applications

The inference rule for function application is

$$\frac{C \vdash e_1 : T_1 \rightarrow T \quad C \vdash e_2 : T_1}{C \vdash e_1 e_2 : T}$$

which can be translated into a Datalog rule

```
typed(X,T) :- app(X,E1,E2), typed(E1,Te), fun(Te,T1,T), typed(E2,T1).
```

In this rule, X is again the name of the complete expression: $x = e_1 e_2$.

3.3.3 Variables

The first approach we tried for variable typing was motivated by the work of [Pacak et al., 2020]. To find the type of a variable, we need to look it up in the context. However, the pitfall here is that introducing contexts allow a single expression to have infinitely many types, as its type inherently depends on the context. To avoid this, the article defines the concept of co-functional dependency: the context is uniquely defined by the expressions in which a term appears. Thus, they created a `context_typed` relation that goes bottom up: starting from a term, we search for its context by looking into the context of the "encompassing" expression. The rules would be of the form

```
context_typed(B,C1) :- lambda(E,V,T,B), context_typed(E,C), lookup(C1,V,T1).
context_typed(E1,C) :- app(E,E1,E2), context_typed(E,C).
context_typed(E2,C) :- app(E,E1,E2), context_typed(E,C).
```

If a term appears in the body of a function, the context to lookup is the context of the function, to which we add the argument of the function and its type. If a term appears on the left hand side of the right hand side of an application, the context is the same as the one of the expression consisting of the application.

This approach works perfectly well, its only inconvenient are its being slow and verbose. We noticed that with our variable unique renaming when parsing the program, we did not need the context anymore. We simplified our Datalog rules by directly assigning the correct type to a variable (as a fact in Datalog), instead of binding it to a context.

3.3.4 Pattern matching

Pattern matching takes a little more work. We divide it into two parts, each being typed by a different inference rule. We first consider each branch of the pattern match independently and try to type it. Once all branches are typed, we check their types are all equal:

$$\frac{C \vdash E : T_1, C \vdash P \rightarrow B : T_1 \rightarrow T}{C \vdash E \text{ match case } P \rightarrow B : T}$$

$$\frac{C \vdash E \text{ match case } P_1 \rightarrow B_1 : T, \dots, C \vdash E \text{ match case } P_n \rightarrow B_n : T}{C \vdash E \text{ match } [\text{case } P_1 \rightarrow B_1, \dots, \text{case } P_n \rightarrow B_n] : T}$$

Their translation into Datalog is

```
typed(X,T) :- match(X,S,P,B), typed(S,T1), typed(P,T1), typed(B,T).
typed(X,T) :- pattern_match_list(X,E1,E2), typed(E1,T), typed(E2,T).
```

where, in `match`, `X` is the name of the complete expression, `S` is the scrutinee, `P` is the pattern, and `B` is the body. In `pattern_match_list`, `X` is also the name of the complete expression, `E1` and `E2` are the identifiers of two branches of the pattern match.

The main difficulty in a pattern match consists in guessing the type of the new introduces variables. Let's consider the function

```
def pred(x: Nat): Nat = {
  x match {
    case z() => z()
    case succ(y) => y
  }
}
```

To typecheck this expression, we need to know the type of the variable `y`, which is defined inside a pattern. We can guess its type by introducing a new relation, `type_constr`, that expressed the type constraints on a variable. We can distinguish two cases:

- The new variable is the pattern itself:

```
x match {
  case y => ...
```

then `y` must have the same type as `x`. We can express this by a general rule

```
type_constr(Y,T) :- match(E,X,Y,B), typed(X,T).
```

- The new variable appears in a constructor (as in the `pred` example where `y` appears in the `succ` constructor). The type constraints must be generated when translating the constructors. For instance, we should add a new rule for natural numbers:

```
type_constr(X,nat) :- succ(E,X).
```

It means that a variable appearing inside the `succ` constructor must be of type `Nat`.

3.4 Represent the AST as Datalog facts

When translating the syntax tree to Datalog for typechecking, each parsed expression is converted into a Datalog fact.

- The variables are given unique labels. When a new variable is created (for example an argument of a function), it creates a new fact,

```
var(x).
```

where `x` is the unique label of the variable. In the rest of Datalog code, the variable is referred to by its label.

- A function application gives rise to a new fact

```
app(app0,f,x)
```

where `app0` is the name of the whole expression, `f` is the function to apply and `x` is the argument given to the function ($app0 = fx$)

- Parsing a function definition creates the facts

```
lambda(lbd0, arg, body).
typed(arg,t).
```

where `lbd0` is the name of the complete expression, `arg` is the argument of the function, and `body` its body. We also provide the types of the arguments.

- For each branch of a pattern match, we have a fact

```
match(match0, scrut, pattern, body).
```

as well as several facts of the form

```
typed(x,T) :- typeconstr(x,T).
```

for each new variable in the pattern.

Finally, the pattern match branches are grouped with

```
pattern_match_list(pat_match0, match1, match2).
```

where `match1` and `match2` are the identifiers of two branches.

3.5 Recursivity

Recursion is a fundamental aspect of computation, and thus we want to ensure that our typechecker works on recursive functions. Since we use top-level bindings instead of the `fix` combinator to build recursivity, we cannot just convert its typing rule into Datalog.

The issue is then that a function only typechecks if all of its inner expressions typecheck as well; however the recursive call will never typecheck because its type will be unknown at this point, by definition:

```
def rec(x: Nat): Nat = { // to type rec...
  rec(x)    // <-- ...we need to type rec
}
```

Here we can see the circular reasoning in action. This gets even more complex for mutually recursive functions, thus we need to find a simple way to break out of the cycle.

The mechanism we propose is simple and relies on the type annotations of the top-level functions. It is comprised of two parts:

- Emit a fact that types a pseudo-function with a similar name:

```
type(tfn1, Nat, Nat). // name the ascribed type
typed(rec_ascribed, tfn1).
```

- Replace all uses of the function `rec` with `rec_ascribed` when called from this or other functions.

Now, recursive functions can typecheck if their type is annotated. The last thing we need is to ensure that the type of function matches its ascribed type:

```
well_typed(rec, T) :- typed(rec, T), typed(rec_ascribed, T).
```

This `well_typed` relation is not emitted currently by our implementation of the typechecker.

3.6 Additional relations

3.6.1 Type equality

As we might generate several names for the same type (for instance `fun(f1,nat,nat)` and `fun(f2,nat,nat)`), we wish to have a notion of type equality. To this end, we introduce a new relation, `type_eq`. We want this relation to be an equivalence relation, so we need to ensure its reflexivity, symmetry and transitivity.

```
type_eq(T,U) :- type(T), T=U.
type_eq(T,U) :- type_eq(U,T).
type_eq(T,U) :- type_eq(T,S), type_eq(S,U).
type_eq(T,U) :- fun(T,T1,T2), fun(U,U1,U2), type_eq(T1,U1), type_eq(T2,U2).
```

The last rules expresses type equality for functions.

3.6.2 Type generation

There is another situation for which we introduce new datalog rules. A term may only be typed by a type that has been defined somewhere with `type(T)`. However, we cannot generate every type because we would need to come up with new names dynamically, which we cannot do in datalog. This is the rule that we would have liked to implement, but cannot:

```
fun(???,T,U) :- type(T), type(U).
```

When typing lambdas, their type may be bigger than any we already have. As an example of the problem in question:

```
def twoargs: Nat -> Nat = {
  \x: Nat. (\b: Bool. x)(true)
} // (\b. x): Bool -> Nat
```

Here, `(\b.x)` (let's call it `lbd_b`) has type `Bool -> Nat`, which is never defined anywhere: there is no fact `fun(???, Bool, Nat)` that could be used to unify with `T` in `typed(lbd_b, T)`.

Hence, we introduce a new relation that borrows the (uniquely generated) names of the lambdas to create a new fun type, where the name is used instead of `???`.

```
fun(X,Tv,Tb) :- lambda(X,V,Tv,B), typed(B,Tb).
```

This allows us to type all lambdas without needing to generate all possible types.

3.7 Example

Here is an example of the transformation from program to Datalog.


```

type Nat {
  case z()
  case s(Nat)
}
def plus(x: Nat, y: Nat): Nat = {
  x match {
    case z() => y
    case s(x1) => s(plus(x1, y))
  }
}

```

```

// Facts (and rules) for the Nat type
type(_Nat0).
typed(_z1,_Nat0).
fun(tfn0,_Nat0,_Nat0).
typed(_s2,tfn0).

typed(E,_Nat0) :- _z1(E).
typed(E,_Nat0) :- _s2(E,A0), typed(A0,_Nat0).

type_constr(X,_Nat0) :- _s2(E,X).

// Facts for the function plus
_z1(pat_conspat4). // z()
match(v5,_x7,pat_conspat4,_y8). // x match case z() => y
var(_x19). // x1 is a variable
typed(_x19,T) :- type_constr(_x19,T). // pattern matching constraint
_s2(pat_conspat6,_x19). // case s(x1)
match(app7,_x7,pat_conspat6,app7). // x match case s(x1) => app7
app(app10,_plus3_ascribed,_x19). // plus(x1)
app(app9,app10,_y8). // (plus(x1))(y)
app(app7,_s2,app9). // s( (plus(x1))(y) )
pattern_match_list(ls_v5,app7,v5). // match list
var(_y8). // y is a var
typed(_y8,_Nat0). // y: Nat
lambda(lbd1,_y8,ls_v5). // \y: Nat. match ...
var(_x7). // x is a var
typed(_x7,_Nat0). // x: Nat
lambda(_plus3,_x7,lbd1). // \x: Nat. \y ...
fun(tfn14,_Nat0,_Nat0).
fun(tfn15,_Nat0,tfn14).
typed(_plus3_ascribed,tfn15). // plus_ascribed: Nat -> Nat -> Nat

typed(_plus3,tfn15)? // does "plus" have type "tfn15"?

```

4 Generic types and polymorphism

If the previous part has been fully automated, we did not have time to automate translation of programs with generic types and polymorphism to Datalog. However, we studied the various Datalog relations needed to typecheck such programs, and tested them "manually" on some examples, such as the identity function or the map function on lists.

4.1 Generic types

The simplest example of a generic type is the list, defined in our languages as follows:

```
type List[A] {  
  case nil()  
  case cons(A, List[A])  
}
```

To translate this data type to Datalog, we need to express the following facts :

- Type definition: if A is a type, then List[A] is a type
- First constructor typing rule: nil[A]() is of type List[A]
- Second constructor typing rule: cons(x,xs) is of type List[A] if x is of type A and xs of type List[A]
- First type constraint for pattern matching: if cons(x,xs) is of type List[A] then x must be of type A
- Second type constraint for pattern matching: if cons(x,xs) is of type List[A] then xs must be of type List[A]

```
type(T) :- type(A), list(T,A).  
typed(E,T) :- nil(E,A), list(T,A).  
typed(E,T) :- cons(E,Elm,List), typed(Elm,A), typed(List,T), list(T,A).  
type_constr(X,T) :- cons(E,X,XS), type_constr(E,T1), list(T1,T).  
type_constr(XS,T) :- cons(E,X,XS), type_constr(E,T).
```

This can be generalized to other data types, using exactly the same principle.

4.2 Polymorphic functions

4.2.1 Challenges

To have a better insight of which kind of challenges arise from polymorphism, let's consider the map function:

```
def map[T, U](f: T -> U, ls: List[T]): List[U] = {  
  ls match {  
    case nil() => nil()  
    case cons(h, t) => cons(f(h), map[T, U](f, t))  
  }  
}
```

- This is a polymorphic function, depending on two polymorphic types T and U
- Its arguments are not directly of types T and U: they are a function and a generic data type, its return type is a generic data type
- The function is recursive
- The constructors nil and cons are not annotated with T and U

The complete example of this function's translation to Datalog is given in the example file `test_lists`, with the corresponding input (with more detailed explanations in `test_lists_input`).

4.2.2 Inference rules for polymorphism

To typecheck polymorphic functions, we need two more inference rules: forall introduction and forall elimination.

$$\frac{C \vdash e : T}{C \vdash \Lambda a.e : \forall a.T} \quad \frac{C \vdash e : \forall a.T}{C \vdash e[U] : [U/a]T}$$

The type $U = \forall a.T$ is expressed by a new relation `type_gen(u,a,t)`.

The Datalog translation of this inference rules is the following:

```
typed(X,U) :- forall_intro(X,E,A), typed(E,T), type_gen(U,A,T),
typed(X,S) :- forall_elim(X,E,A,U), typed(E,T1), type_gen(T1,A,T),
               replace(S,U,A,T).
```

where `replace(S,U,A,T)` is the relation that expresses $S = [U/a]T$

4.2.3 The replace relation

For the forall elimination rule, we need to define the `replace` relation, that replaces a polymorphic type by its actual type. We define in Datalog the equivalent of a recursive function:

- If the type is the polymorphic type, we replace it by the actual type
- If we encounter a non-generic data type (such as Bool or Nat) we do nothing
- If we encounter a function type, we continue replacing in the argument type and the return type
- If we encounter the type $\forall b.T$, we continue replacing in T.

```
replace(T,U,A,T1) :- type_eq(A,T1), type_eq(T,U).
replace(T,U,A,T1) :- simple_type(T1), T=T1, type(U), type(A).
replace(T,U,A,T1) :- fun(T1,T2,T3), replace(R2,U,A,T2), replace(R3,U,A,T3),
               fun(T,R2,R3).
replace(T,U,A,T1) :- type_gen(T1,B,T2), replace(R2,U,A,T2), type_gen(T,B,R2).
```

When defining generic data types such as lists, we need to add a new rule to the type definition: if we want to call the `replace` function on a type `List[A]`, we need to call the `replace` function on A:

```
replace(T,U,A,T1) :- list(T1,T2), replace(R2,U,A,T2), list(T,R2).
```

4.3 Example : the identity function

We wish to correctly typecheck the (polymorphic) identity function, as well as its applications on arguments of type Bool and Nat.

To define the identity function, we go through the following steps:

- Define the function variable with a unique fresh name (here x)
- Type this variable. Here, its type is polymorphic, so give it a fresh name (here t)
- Give a name to the type $t \rightarrow t$ (here `t_t`)
- Define the function $\lambda x.x$ and give it a name (here `id`)
- As this function is polymorphic, we need a forall introduction (called `forall_id` here)

- Give a name to the new type $\forall t.t \rightarrow t$ (here `id_type`)

```
var(x).
typed(x,t).
fun(t_t,t,t).
lambda(id,x,x).
forall_intro(forall_id,id,t).
type_gen(id_type,t,t_t).
```

We can then apply this function to a natural number (for example zero): we first call the `forall` elimination rule to replace the polymorphic type `t` by the type `Nat`, and we apply the result to zero:

```
forall_elim(id_nat,forall_id,t,T) :- typed(zero,T).
app(id_zero,id_nat,zero).
```

We can do the same for booleans:

```
forall_elim(id_bool,forall_id,t,T) :- typed(_true,T).
app(id_true,id_bool,_true).
```

Here are the results we obtain for some queries:

```
typed(id,T)?
> typed(id, t_t).
typed(forall_id,T)?
> typed(forall_id, id_type).
typed(id_bool,T)?
> typed(id_bool, funboolbool).
typed(id_nat,T)?
> typed(id_nat, funnatnat).
typed(id_zero,T)?
> typed(id_zero, nat).
typed(id_true,T)?
> typed(id_true, bool).
```

The whole example can be found in the file `test_identity`.

5 Limitations

After implementing our system, we found that it did exhibit a number of limitations:

- The type checking is very slow. It is probably due in part to our choice of interpreter, which is very basic and thus probably doesn't try to optimize the searches. The way we generate datalog clauses is also quite naïve and tries to be as simple as possible, even if the run-time suffers.
- Using datalog complicates the production of error messages. Indeed, our current approach queries Datalog to check whether the system finds a valid type for all terms, but there is no way to ask it why it failed if it did, i.e. which sub-terms could not be typed and which were the mismatching types.
- Since we cannot generate an infinite amount of types, there is no clear path towards building a typechecker for polymorphic types in Datalog. There could be a way to generate more types by using some of the non-Datalog features of Soufflé. We can generate new names by using the

pseudo-relation `cat`, which concatenates the string representation of the atoms given to it, which allows us to generate an infinite amount of types. We also limit the number of elements of this type that we generate using a `.limitsize` directive to prevent Soufflé from looping forever.

```
type(F) :- type(A), type(B), fun(F,A,B).  
fun(cat("(",cat(cat(A,"->"),cat(B,")"))),A,B) :- type(A), type(B).  
.limitsize fun(n=100) // 100 is arbitrary
```

The issue with this workaround is that some valid programs may not typecheck because the limit prevents the solver from instantiating a big enough type: we lose the fact that Datalog is supposed to be complete, namely that it does deduce every true fact.

6 Conclusion and further work

We showed that using Datalog to type programs is feasible for simple type systems, although it presents a number of challenges. The idea is elegant in theory, but doesn't adapt itself very well to the other concrete constraints of typechecking, such as giving error messages and performance. Our implementation is also restrained to simple type systems where the complexity benefits of using Datalog are less important. However, we show that in spite of the other issues it is still possible to type programs with Datalog. A path towards improvement would be to explore whether we can perform type inference as well, at least for simple type systems.

Accessing the code

You can find our code at the following git repository:

<https://github.com/smolene/formal-verification-project>

References

- [1] André Pacak, Sebastian Erdweg, and Tamas Szabo. A systematic approach to deriving incremental type checkers. *Proceedings of the ACM on Programming Languages*, 4:1–28, 11 2020.